

## PROCESAMIENTO AVANZADO DE PANTALLA

**Objetivos:** estudiar características avanzadas del manejo de la pantalla, como recorrido, video inverso, intermitencia y despliegue de video en memoria.

### AREA DE DESPLIEGUE DE VIDEO EN MEMORIA

Esta área de memoria contiene la información que el monitor va a desplegar. Las interrupciones del DOS y BIOS, manejan el despliegue en pantalla en forma directa, transfiriendo los datos a esta área, sin que tengamos que dar la dirección física de memoria.

Estas son las direcciones del inicio de los segmentos de despliegue de video:

- A000:[0] Utilizada para gráficos de alta resolución
- B000:[0] Modo texto monocromático
- B100:[0] Para tarjeta de gráficos Hércules (fuera de uso)
- **B800:[0]** Modo texto a color para monitor gráfico de color RGB (Red-Green-Blue)

Esta es el área que nos interesa

### MODOS DE TEXTO

El modo texto se utiliza para el despliegue normal en la pantalla de caracteres ASCII. El procesamiento es similar tanto para monocromático, como para color.

Modo que utilizaremos usualmente

- |  |  |
|--|--|
| Modo texto 00 (mono) y 01 (color)        | : formato de 40 columnas   |
| Modo texto 02 (mono) y <b>03 (color)</b> | : formato de 80 columnas   |
| Modo texto 07 (mono)                     | : para MDA (Adaptador de pantalla monocromática), EGA (Adaptador de gráficos mejorado), VGA (Matriz de gráficos de video). |

La función 00H de la Interrupción 10H del BIOS, selecciona el modo de video. Veamos un ejemplo de cómo se utiliza:


```
MOV AH, 00H ; petición para designar el modo
MOV AL, 03h ; modo de texto o estándar a color 80x25
INT 10H ; llama al BIOS
```

## BYTE DE ATRIBUTO

Es el byte que determina las características de cada carácter mostrado en modo *texto*.

Cuando se establece un atributo, este permanece activado hasta que otra operación lo cambie, por tanto los caracteres desplegados tendrán el mismo atributo.

Formato del byte de atributo:



Las letras R, G, B indican las posiciones de bits para Rojo, Verde y Azul.

Bit 7 (BL) : Establece intermitencia  
 Bit 6-4 : Determina fondo de la pantalla  
 Bit 3 (I) : Establece intensidad alta  
 Bit 2-0 : Determina el color del carácter desplegado (frente)

Ejemplos de algunos atributos comunes:

Fondo	Primer plano	Fondo BL R G B	Primer plano I R G B	Hex
Negro	Azul	0 0 0 0	0 0 0 1	01
Verde	Azul	0 0 1 0	0 0 1 1	23
Verde	Gris intermitente	1 0 1 0	1 0 0 0	A8

## PAGINAS DE PANTALLA

- Los modos de texto permiten almacenar datos en memoria de video en *páginas*.
- Las páginas van de 0 hasta 3 en modo normal de 80 columnas.

Página 0: B800H:[0]  
Página 1: B900H:[0]  
Página 2: BA00H:[0]  
Página 3: BB00H:[0]

- Pueden formatearse todas las páginas en memoria, pero solo puede desplegar una a la vez.
- Cada carácter que se muestra en pantalla necesita dos bytes de memoria: uno para el carácter y el segundo para el atributo.
- Cada página ocupa en memoria realmente 4,000 bytes (80 col x 25 filas x 2 bytes, aunque no necesariamente se use por completo).
- La **función 02H de la Int 10H del BIOS** permite ubicar el cursor en la pantalla de acuerdo a la coordenada fila, columna, en la página que se seleccione:

BH = Número de página para modo texto con 80 columnas. (0, 1, 2, 3)

DH = fila

DL = Columna

MOV AH, 02H	; Petición
MOV BH, 00H	; Página 0
MOV DH, 05	; Renglón
MOV DL, 20	; Columna
INT 10H	

- La función 05H de la Int 10H del BIOS permite seleccionar la página activa:

MOV AH, 05H	; Petición
MOV AL, PAGINA	; Número de página
INT 10H	

- La función 13H de la Int 10H del BIOS permite desplegar una cadena de caracteres, con opción de establecer el atributo. Los registros ES:BP deben contener la dirección segmento:desplazamiento de la cadena que se despliega.

MOV AH, 13H	; Petición
MOV AL, subfuncion	;
MOV BH, pagina	; 0, 1, 2, 3
MOV BL, atributo	; atributos de pantalla
MOV BP, dirección	; dirección de la cadena en ES:BP
MOV CX, longitud	; longitud de la cadena de caracteres
MOV DX, pantalla	; posición relativa de inicio en la pantalla
INT 10H	

Subfunciones del AL:

- 00 Despliega atributo y cadena, no avanza el cursor
- 01 Despliega atributo y cadena, si avanza el cursor
- 02 Despliega carácter y después atributo, no avanza el cursor
- 03 Despliega carácter y después atributo, si avanza el cursor

## DESPLIEGUE DIRECTO EN VIDEO

- Se envían los caracteres directamente a la memoria sin utilizar el DOS ni el BIOS.
- La primera página (0) inicia en B800:[0]
- La posición 0,0 equivale a B800:0000
- La posición 0, 1 equivale a B800:0002, puesto que cada carácter en pantalla necesita dos bytes de memoria, uno para carácter y el que le sigue para su atributo

Muy importante

El siguiente ejemplo muestra una serie de caracteres, utilizando el despliegue directo en video:

```
.model small
.stack 64
;-----
.data
mensaje DB 'Prueba de despliegue directo en video', '$'
;-----
.code
    mov ax, @DATA      ; inicializar area de datos
    mov ds, ax

    lea dx, mensaje    ; mensaje inicial
    mov ah, 09h
    int 21h

    mov ax, 0B800h ; dirección de inicio de memoria de video
                  ; pagina 0
    mov es, ax      ; se carga al registro ES la direcc mem ¡MUY IMPORTANTE!

    mov di, 0
    mov cx, 1700    ; numero de veces que se repite la impresion
    mov al, 42h     ; se carga el caracter que va a imprimirse (B)
    mov ah, 5       ; atributo
    rep stosw       ; almacena el contenido del acumulador en
                  ; la memoria. El prefijo REP junto con CX
                  ; hace que se repita la operacion CX veces
                  ; El par ES:DI hace referencia al area de
                  ; memoria donde sera almacenado

    mov ax, 4C00H
    int 21h
```

- A continuación veremos la sintaxis de las instrucciones STOSW y REP, que son utilizadas en este programa:

- STOS/STOSB/STOSW (almacena cadena)

Sintaxis:   STOS     cadena\_destino  
               STOSB    (bytes)  
               STOSW    (palabras)

- Transfiere el operando origen almacenado en AX o AL, al destino direccionado por el registro DI sobre el segmento extra (ES).
- Tras la operación, DI se incrementa o decrementa según el indicador DF (véanse CLD y STD) para apuntar al siguiente elemento de la cadena.
- *Cadena\_destino* es un operando redundante que sólo indica el tipo del dato (byte o palabra) a cargar. Es más cómodo colocar STOSB o STOSW para indicar bytes/palabras.

Ejemplo:   lea     di,destino  
               mov    ax,1991  
               stosw

- REP (repetir cadena de caracteres)

Sintaxis: REP instrucción-de-cadena

- Repite una operación de cadena un número específico de veces.
- Es un prefijo de repetición opcional codificado antes de las instrucciones de cadena MOVS y STOS.
- Carga el CX con un contador antes de la ejecución.
- Para cada ejecución de la instrucción de cadena, REP disminuye en 1 y repite la operación hasta que CX es cero

El siguiente ejemplo muestra una cadena de caracteres definida en el área de datos del programa, utilizando el despliegue directo en video:

```

;-----
; Despliegue de un mensaje del area de datos
; a la memoria de video
;-----
.model small
.stack 64
.data
;-----
mensaje    db    'Esta es una prueba'
cont       db    0
;-----
.code
    mov ax, @DATA ; inicializar area de datos
    mov ds, ax

    mov ax, 0B800h ; direccion de inicio de memoria de video
    mov es, ax     ; se carga al registro ES la dir direcc mem
    mov di, 0
    mov cont, 0    ; inicializa contador a cero
    lea bx, mensaje ; mensaje se carga en bx
ciclo:
    mov al, [bx]   ; se carga el caracter que va a imprimirse
    mov ah, 5      ; atributo
    stosw          ; almacena el contenido del acumulador en
                  ; la memoria.
                  ; El par ES:DI hace referencia al area de
                  ; memoria donde sera almacenado
    inc bx         ; incrementa puntero de la cadena
    inc cont       ; incrementa contador
    cmp cont, 18   ; hasta llegar al tamaño de la cadena
    jne ciclo

    mov ax, 4C00H ; salir a DOS
    int 21h
end

```